Import data from Access 2007 database to Excel 2007
A common task is to import data with VBA to Excel from Access databases.  Access 2007 has undergone some major important changes which effect on how we create connections to Access 2007 databases and also how we work with Access 2007 objects model.  
To communicate with Access 2007 databases via VBA the preferable method is to use Data Access Objects (DAO) which is the object model written specifically for Access.  It’s an object library with a collection of database objects.
The first step is to set a reference to the new library Microsoft Office 12 Access Database Engine Object Library via the command Tools | References… in the VB-editor in Excel 2007.
The general process is:
· Connecting to the database
· Send a SQL (Structured Query Language) question in order to populate a recordset with the wanted records. A recordset can be considered as a ‘container’ which contain records and fields where its contents
· Is copied to a worksheet and finally
· Closing the connection to the database
The following code example shows a number of possibly approaches to retrieve data in flexibly ways:
· Retrieve whole table’s data 
· Retrieve a selected group of data based on SQL queries
· Execute stored reports in the database
· Retrieve a selected group of data based on parameterized saved queries 

Retrieve a whole table’s data
This should only be done if it’s necessary and the table don’t contain a large number of records. In the example a table-type recordset is used and only a table name can be the source.

Retrieve a selected group of data based on SQL queries
This is the most common approach in that we create a SQL queries which include specific WHERE clause in order to retrieve the wanted records.  In the example we use the ‘snapshot’ recordset type. A snapshot-type recordset is static, i e it’s not editable and therefore it can only be read. 

SQL is a powerful language and we can create very complex questions but it’s beyond this hack to delve into it deeper. 
Execute a stored report in the database
Instead of creating the SQL question in VBA we can create and save the question in the Access 2007 database and then execute it from Excel 2007. Here we use the forward-type recordset type which is efficient and allow us to move forward in the recordset (if wanted).
Retrieve a selected group of data based on a parameterized saved queries
This is also a common way to retrieve the wanted data where the parameters express certains conditions in terms of WHERE-clauses. Per se a recordset that include parameters is of the type dynaset which is the most flexible type of them all, editable and can scroll forward and backwards in the recordset.
Option Explicit
'A reference via Tools | References... to Microsoft Office 12 Access Database Engine Object Library
'must be set in order to get the below code to work properly.

Sub Import_Data_Access_2007()

'The pathway to and the name of the database.
Const stDB As String = "c:\Northwind 2007.accdb"

'The table to be retrieved all data from.
Const stWholeTable As String = "Shippers"

'The SQL query with one condition.
Const stSQL As String = "SELECT Company,City FROM Shippers " & _
                        "WHERE [Country/Region]='USA'"

'The name of the stored report that will be executed.
Const stStoredReport As String = "Order Summary"

'The parameterized stored query that will be executed.
Const stStoredQuery As String = "qryEmplLoc"
                   
'A general error handling routine.
On Error GoTo Error_Handling

'Variables for the DAO objects.
'The database object.
Dim db As DAO.Database
'The table defintion variable for a saved table object.
Dim tdf As DAO.TableDef
'The saved query object.
Dim qdf As DAO.QueryDef
'The recordset object.
Dim rs As DAO.Recordset

'Variables for the Excel objects.
Dim wbTarget As Workbook
Dim wsOrders As Worksheet
Dim wsShippers As Worksheet
Dim wsProducts As Worksheet
Dim wsEmployees As Worksheet
Dim rnOrders As Range
Dim rnShippers As Range
Dim rnProducts As Range
Dim rnEmployees As Range
Dim lnCounter As Long

'Instantiate the Excel objects.
Set wbTarget = ActiveWorkbook

With wbTarget
    Set wsOrders = .Worksheets(1)
    Set wsShippers = .Worksheets(2)
    Set wsProducts = .Worksheets(3)
    Set wsEmployees = .Worksheets(4)
End With

Set rnOrders = wsOrders.Range("A2")
Set rnShippers = wsShippers.Range("A2")
Set rnProducts = wsProducts.Range("A2")
Set rnEmployees = wsEmployees.Range("A2")

'Instantiate the DAO objects and at the same time create
'a connection to the database and populate the 1st recordset.
Set db = OpenDatabase(stDB)
Set tdf = db.TableDefs(stWholeTable)
Set rs = tdf.OpenRecordset(dbOpenTable)

'To avoid that the screen flicker during dataprocessing.
Application.ScreenUpdating = False

'Populate the first row in the target worksheet with fieldnames.
For lnCounter = 0 To rs.Fields.Count - 1
    'The array of fieldnames is 0-based which we need to consider when
    'populating the worksheet.
    wsShippers.Cells(1, lnCounter + 1).Value = rs.Fields(lnCounter).Name
Next lnCounter

'Copy the retrieved records.
rnShippers.CopyFromRecordset rs

'Clear the recordset variable.
Set rs = Nothing
'Instantiate and populate the 2nd recordset.
Set rs = db.OpenRecordset(stSQL, dbOpenSnapshot)

'Populate the first row in the target worksheet with fieldnames.
For lnCounter = 0 To rs.Fields.Count - 1
    'The array of fieldnames is 0-based which we need to consider when
    'populating the worksheet.
    wsProducts.Cells(1, lnCounter + 1).Value = rs.Fields(lnCounter).Name
Next lnCounter

'Copy the retrieved records.
rnProducts.CopyFromRecordset rs

'Clear the recordset variable.
Set rs = Nothing
'Instantiate and populate the 3rd recordset.
Set rs = db.OpenRecordset(stStoredReport, dbOpenForwardOnly)

'Populate the first row in the target worksheet with fieldnames.
For lnCounter = 0 To rs.Fields.Count - 1
    'The array of fieldnames is 0-based which we need to consider when
    'populating the worksheet.
    wsOrders.Cells(1, lnCounter + 1).Value = rs.Fields(lnCounter).Name
Next lnCounter

'Copy the retrieved records.
rnOrders.CopyFromRecordset rs

'Clear the recordset variable.
Set rs = Nothing

'Instantiate the QueryDef object.
Set qdf = db.QueryDefs(stStoredQuery)
'Add a value for the parameter ZIP Code.
qdf.Parameters("[ZIP Code]") = "99999"
'Instantiate and populate the 4th recordset.
Set rs = qdf.OpenRecordset(dbOpenDynaset)

'Populate the first row in the target worksheet with fieldnames.
For lnCounter = 0 To rs.Fields.Count - 1
    'The array of fieldnames is 0-based which we need to consider when
    'populating the worksheet.
    wsEmployees.Cells(1, lnCounter + 1).Value = rs.Fields(lnCounter).Name
Next lnCounter

'Copy the retrieved records.
rnEmployees.CopyFromRecordset rs

'Close the objects.
rs.Close
db.Close

MsgBox "All data has been successfully transfered.", vbOKOnly

'Make sure that it only exist one exit point in the procedure.
ExitSub:
'Release the objects from memory.
Set rs = Nothing
Set qdf = Nothing
Set tdf = Nothing
Set db = Nothing
Exit Sub

Error_Handling:
MsgBox "Error number: " & Err.Number & vbNewLine & _
       "Description: " & Err.Description, vbOKOnly
Resume ExitSub

End Sub 








